SIMULATION MODEL OF THE INSURANCE COMPANY

U.I. Behunkou, S.S. Smorodinsky
Minsk, Belarus
e-mail: vladimir_begunkov@yahoo.com

Abstract

The purpose of this study is to describe a problem of starting new insurance business and specify a work process of a typical insurance company. This article also shows how a connection between Delphi and GPSS World can be used for insurance company simulation and obtaining its profit over a certain period.

1 Introduction

Starting new insurance business is very risky nowadays. It is very important to know whether business is successful and profitable. But nobody can predict the results without detailed analysis, especially when the success of the enterprise depends on many parameters. For solving such issues simulation modeling is widely used. Therefore in this article a simulation model of a insurance company is represented. This model can help to predict a profit of an enterprise over a certain period and establish development strategy.

2 Problem formulation

It is necessary to simulate a work processes of the insurance company which concludes two types of insurance contract and calculates its profit over a certain period. A simulation period can be from 1 day to 360 days. The insurance company has two types of contracts. Under these contracts claims are made with intensities which represent piecewise constant functions and have constant values during the first half of the year and the others constant values during the second half of the year; so the seasonal fluctuations are taken into account. Consequently, the intensities of the first type incoming claims are μ_{11}, μ_{12} for the first and second half of the year respectively and the intensities of the second type incoming claims are μ_{21}, μ_{22} for the first and second half of the year respectively. At first incoming claims are assigned to adjusters who evaluate them with μ_1 and μ_2 intensities. The first and second type claims are processed by m_1 and m_2 adjusters respectively. Then a cashier pays claims with μ_v intensity. After that claims considered processed. But if the insurance event occurs again the process repeats. It continues before the end of the contract term. So a closed queuing net can be used as a model of the described insurance company. Also it is needed to calculate daily and aggregated profit of the company. For calculation of daily profit the formula below is used [3]:

$$P_d = D_o * (K - K_{ev1} + K_{ev2}) - \sum_{i=1}^{2} D_i * K_{evi} - D_v * K_v - \sum_{i=1}^{2} E_i * m_i - F$$

P_d - company daily profit; D_o - company daily income from one contract when the policyholder don't make a claim; K - sum of K_1 and K_2, where K_1 and K_2 are contracts of the first and second type respectively; K_{ev1} - number of the first type claims at an evaluation stage; K_{ev2} - number of the second type claims at an evaluation stage; D_1 - daily company losses from the first type claim being at an evaluation stage; D_2 - daily company losses from the second type claim being at an evaluation stage; D_v - company losses from any type claim being paid; K_v - number of claims being at a payment stage; E_1 and E_2 - daily rates of the adjusters who evaluate the first and second type claims respectively; F - cashier daily rate. An aggregated profit is calculated as sum of all daily profits.

3 Program description

The program consists of two parts: a client application and a simulation model. The client application has been developed with Delphi language. It contains two forms. The first one is used for entering insurance company characteristics: K_1, K_2, μ_{11}, μ_{12}, μ_{21}, μ_{22}, m_1, m_2, μ_1, μ_2, μ_v, D_o, D_1, D_2, D_v, E_1, E_2, F and a simulation period. These parameters are used in the simulation process. For simulation it is necessary to fill up the first form with the characteristics and push the "Simulate" button as shown in the figure 1.



Figure 1: Simulation Characteristics

Then Delphi program saves all characteristics in the text files and launches GPSS simulation. The simulation model reads all characteristics from the text files and uses them for simulation. After simulation all result data, daily profits and accumulated profit over the period, are also saved in the text files. Then Delphi program reads all these data and launches the second form. This form represents daily profits via chart and aggregated profit in the separate box as shown in the figure 2.



Figure 2: Simulation Results

4 Simulation model

This simulation model was developed with GPSS World [1, 2].

```
*files opening for writing and reading, reading the characteristics
        generate ,,,1
        open ("data.txt"),1,Failure_Open
        open ("result1.txt"),2,Failure_Open
        open ("result2.txt"),3,Failure_Open
        read parameter,1,Failure_Read
        savevalue number_contracts1,p$parameter
        read parameter,1,Failure_Read
        savevalue number_employee1,p$parameter
        read parameter,1,Failure_Read
        savevalue intension_claim1_winter,p$parameter
        read parameter,1,Failure_Read
        savevalue intension_claim1_summer,p$parameter
        read parameter,1,Failure_Read
        savevalue intension_processing1,p$parameter
        read parameter,1,Failure_Read
        savevalue number_contracts2,p$parameter
        read parameter,1,Failure_Read
        savevalue number_employee2,p$parameter
        read parameter,1,Failure_Read
        savevalue intension_claim2_winter,p$parameter
        read parameter,1,Failure_Read
        savevalue intension_claim2_summer,p$parameter
        read parameter,1,Failure_Read
```

```
        savevalue intension_processing2,p$parameter
        read parameter,1,Failure_Read
        savevalue intension_payment,p$parameter
        read parameter,1,Failure_Read
        savevalue period_modeling,p$parameter
*reading of the economic characteristics
        read parameter,1,Failure_Read
        savevalue client_income,p$parameter
        read parameter,1,Failure_Read
        savevalue loss_type1,p$parameter
        read parameter,1,Failure_Read
        savevalue loss_type2,p$parameter
        read parameter,1,Failure_Read
        savevalue loss_payment,p$parameter
        read parameter,1,Failure_Read
        savevalue maintenance_cashier_costs,p$parameter
        read parameter,1,Failure_Read
        savevalue maintenance_appraiser_costs1,p$parameter
        read parameter,1,Failure_Read
        savevalue maintenance_appraiser_costs2,p$parameter
*first data stream closing
        close Prob,1,Failure_Close
*creation of active transaction copies
        split x$number_contracts1,first_type
        split x$number_contracts2,second_type
        split 1,start_modeling
        terminate
*multichannel facilities declaration
        include "data2.txt"
*stream of the first type claims
first_type       assign 1,1
tipe1            savevalue number_type_1+,1
                 queue line_type1
                 seize one
                 depart line_type1
                 advance (exponential(1,0,x$intension_claim1_winter))
                 release one
                 transfer ,declare1
*stream of the second type claims
second_type     assign 1,2
tipe2           savevalue number_type_2+,1
                queue line_type2
                seize two
                depart line_type2
```

```
                advance (exponential(2,0,X$intension_claim2_winter))
                release two
                transfer ,declare2
*claims processing
*processing of the first type claims
declare1  savevalue number_type_1-,1
          queue line1
          savevalue processing_type_1+,1
          enter processing1
          depart line1
          advance x$intension_processing1
          leave processing1
          savevalue processing_type_1-,1
          transfer ,payments
*processing of the second type claims
declare2  savevalue number_type_2-,1
          queue line2
          savevalue processing_type_2+,1
          enter processing2
          depart line2
          advance x$intension_processing2
          leave processing2
          savevalue processing_type_2-,1
*claims payments
payments  queue line_payment
          savevalue payment_all_type+,1
          seize payment
          depart line_payment
          advance x$intension_payment
          release payment
          savevalue payment_all_type-,1
*type checking
          test e p1,1,tipe2
          transfer ,tipe1
*changing the intensity of the claims streams
          generate ,,,1
          advance 180
          savevalue intension_claim1_winter,x$intension_claim1_summer
          savevalue intension_claim2_winter,x$intension_claim2_summer
          terminate
*company profit  calculation
          generate ,,,1
new_day   advance 1
          savevalue day_profit,(x$client_income#
```

```
          (x$number_contracts1+x$number_contracts2-x$processing_type_1-
          x$processing_type_2-x$payment_all_type)-x$loss_type1#
          x$processing_type_1-x$loss_type2#x$processing_type_2-x$loss_payment#
          x$payment_all_type-x$maintenance_appraiser_costs1#x$number_employee1-
          x$maintenance_appraiser_costs2#x$number_employee2-
          x$maintenance_cashier_costs)
          write x$day_profit,2,Failure_Write,off
          savevalue profit+,x$day_profit
          transfer ,new_day
          write x$day_profit,2,Failure_Write,off
          savevalue profit+,x$day_profit
          transfer ,new_day
*simulation period
start_modeling   advance x$period_modeling
                 write x$profit,3,Failure_Write,off
                 close Prob2,2,Failure_Close
                 close Prob3,3,Failure_Close
                 terminate 1
*files closing/opening and data writing/reading errors handling
Failure_Open terminate 1
Failure_Read terminate 1
Failure_Close terminate 1
Failure_Write terminate 1

start 1
exit -1
```

The operator 'generate ,,,1' creates one transact which goes to the sequence of the blocks 'open ("data.txt"),1,Failure_Open ... close Prob,1,Failure_Close'. These blocks open two data streams, read the characteristics from the first text file, save them as parameters and then close the first data stream. In case of unsuccessful opening/closing any data stream or writing/reading data from the files, the active transaction goes to the 'Failure_Open terminate 1 ... Failure_Write terminate 1' block and the simulation ends. In the other case the active transaction goes to 'split x$number_contracts1,first_type ... split 1,start_modeling' block. In this block copies of the transaction are created for simulation of the first and second type claims. These transactions go to 'first_type' and 'second_type' labels. This block also creates one copy of the transaction which goes to 'start_modeling advance x$period_modeling' block and starts the period of simulation. Next operator 'include "data2.txt"' inserts into the model a storages declaration from the file. The content of this file is shown below. The block 'first_type assign 1,1 ... transfer ,declare2' simulates claims incoming. Then the block 'declare1 savevalue number_type_1-,1 ... savevalue processing_type_2-,1' simulates claims evaluation by adjusters. Next block 'payments queue line_payment... savevalue payment_all_type-,1' simulates claims payments. The operators 'test e p1,1,tipe2 transfer ,tipe1' check a type

of the transaction. Depending on the result the transaction goes to 'type1' or 'type2' labels. This means that the claim is not processed before next insurance event. Following block 'generate ,,,1 ... savevalue intension_claim2_winter,x$intension_claim2_summer' simulates changing of the claims streams intensities. The block 'generate ,,,1 ... transfer ,new_day' calculates daily and aggregated profit. It also writes daily profit to the result file. Next block 'start_modeling ... terminate 1' controls the period of the simulation. At the end of the period the aggregated profit is written to the results file, the second and third data streams are closed and the simulation ends.

Content of the "data2.txt" file:

```
processing1 storage 2
processing2 storage 3
```

References

[1] Elina-Computer (2002). *GPSS World Tutorial Manual*. "Master-Line", Kazan.

[2] Elina-Computer (2002). *GPSS World Reference Manual*. "Master-Line", Kazan.

[3] Matalytski M.A., Romaniuk T.V. (2005). Mathematical analysis of the stochastic models for multi-type claim processing in insurance companies. *Doklady of the National Academy of Sciences of Belarus*. Vol. **49**, No.1, pp. 18-23.